

# NAG Library Routine Document

## F08KBF (DGESVD)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

### 1 Purpose

F08KBF (DGESVD) computes the singular value decomposition (SVD) of a real  $m$  by  $n$  matrix  $A$ , optionally computing the left and/or right singular vectors.

### 2 Specification

```
SUBROUTINE F08KBF (JOBU, JOBVT, M, N, A, LDA, S, U, LDU, VT, LDVT, WORK,      &
                  LWORK, INFO)
INTEGER          M, N, LDA, LDU, LDVT, LWORK, INFO
REAL (KIND=nag_wp) A(LDA,*), S(*), U(LDU,*), VT(LDVT,*),      &
                  WORK(max(1,LWORK))
CHARACTER(1)     JOBU, JOBVT
```

The routine may be called by its LAPACK name *dgesvd*.

### 3 Description

The SVD is written as

$$A = U\Sigma V^T,$$

where  $\Sigma$  is an  $m$  by  $n$  matrix which is zero except for its  $\min(m,n)$  diagonal elements,  $U$  is an  $m$  by  $m$  orthogonal matrix, and  $V$  is an  $n$  by  $n$  orthogonal matrix. The diagonal elements of  $\Sigma$  are the singular values of  $A$ ; they are real and non-negative, and are returned in descending order. The first  $\min(m,n)$  columns of  $U$  and  $V$  are the left and right singular vectors of  $A$ .

Note that the routine returns  $V^T$ , not  $V$ .

### 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

### 5 Arguments

- |  |   |
|--|---|
| 1: <b>JOBU</b> – CHARACTER(1)  | <i>Input</i>  |
| <i>On entry:</i> specifies options for computing all or part of the matrix $U$ . |   |
| JOBU = 'A'   | All $m$ columns of $U$ are returned in array $U$ .  |
| JOBU = 'S'   | The first $\min(m,n)$ columns of $U$ (the left singular vectors) are returned in the array $U$ .    |
| JOBU = 'O'   | The first $\min(m,n)$ columns of $U$ (the left singular vectors) are overwritten on the array $A$ . |

**JOBU = 'N'**

No columns of  $U$  (no left singular vectors) are computed.

*Constraint:*  $\text{JOBU} = \text{'A'}$ ,  $\text{'S'}$ ,  $\text{'O'}$  or  $\text{'N'}$ .

2:  $\text{JOBVT} - \text{CHARACTER}(1)$

*Input*

*On entry:* specifies options for computing all or part of the matrix  $V^T$ .

**JOBVT = 'A'**

All  $n$  rows of  $V^T$  are returned in the array VT.

**JOBVT = 'S'**

The first  $\min(m, n)$  rows of  $V^T$  (the right singular vectors) are returned in the array VT.

**JOBVT = 'O'**

The first  $\min(m, n)$  rows of  $V^T$  (the right singular vectors) are overwritten on the array A.

**JOBVT = 'N'**

No rows of  $V^T$  (no right singular vectors) are computed.

*Constraints:*

$\text{JOBVT} = \text{'A'}$ ,  $\text{'S'}$ ,  $\text{'O'}$  or  $\text{'N'}$ ;

$\text{JOBVT}$  and  $\text{JOBU}$  cannot both be ' $O$ '.

3:  $M - \text{INTEGER}$

*Input*

*On entry:*  $m$ , the number of rows of the matrix  $A$ .

*Constraint:*  $M \geq 0$ .

4:  $N - \text{INTEGER}$

*Input*

*On entry:*  $n$ , the number of columns of the matrix  $A$ .

*Constraint:*  $N \geq 0$ .

5:  $A(\text{LDA},*) - \text{REAL} (\text{KIND}=\text{nag\_wp}) \text{ array}$

*Input/Output*

**Note:** the second dimension of the array A must be at least  $\max(1, N)$ .

*On entry:* the  $m$  by  $n$  matrix  $A$ .

*On exit:* if  $\text{JOBU} = \text{'O'}$ , A is overwritten with the first  $\min(m, n)$  columns of  $U$  (the left singular vectors, stored column-wise).

If  $\text{JOBVT} = \text{'O'}$ , A is overwritten with the first  $\min(m, n)$  rows of  $V^T$  (the right singular vectors, stored row-wise).

If  $\text{JOBU} \neq \text{'O'}$  and  $\text{JOBVT} \neq \text{'O'}$ , the contents of A are destroyed.

6:  $\text{LDA} - \text{INTEGER}$

*Input*

*On entry:* the first dimension of the array A as declared in the (sub)program from which F08KBF (DGESVD) is called.

*Constraint:*  $\text{LDA} \geq \max(1, M)$ .

7:  $S(*) - \text{REAL} (\text{KIND}=\text{nag\_wp}) \text{ array}$

*Output*

**Note:** the dimension of the array S must be at least  $\max(1, \min(M, N))$ .

*On exit:* the singular values of  $A$ , sorted so that  $S(i) \geq S(i + 1)$ .

8:	$U(LDU, *)$ – REAL (KIND=nag_wp) array	<i>Output</i>
<b>Note:</b> the second dimension of the array $U$ must be at least $\max(1, M)$ if $\text{JOB}U = 'A'$ , $\max(1, \min(M, N))$ if $\text{JOB}U = 'S'$ , and at least 1 otherwise.		
<i>On exit:</i> if $\text{JOB}U = 'A'$ , $U$ contains the $m$ by $m$ orthogonal matrix $U$ .		
If $\text{JOB}U = 'S'$ , $U$ contains the first $\min(m, n)$ columns of $U$ (the left singular vectors, stored column-wise).		
If $\text{JOB}U = 'N'$ or ' $O$ ', $U$ is not referenced.		
9:	$LDU$ – INTEGER	<i>Input</i>
<i>On entry:</i> the first dimension of the array $U$ as declared in the (sub)program from which F08KBF (DGESVD) is called.		
<i>Constraints:</i>		
if $\text{JOB}U = 'A'$ or ' $S$ ', $LDU \geq \max(1, M)$ ; otherwise $LDU \geq 1$ .		
10:	$VT(LDVT, *)$ – REAL (KIND=nag_wp) array	<i>Output</i>
<b>Note:</b> the second dimension of the array $VT$ must be at least $\max(1, N)$ if $\text{JOB}VT = 'A'$ or ' $S$ ', and at least 1 otherwise.		
<i>On exit:</i> if $\text{JOB}VT = 'A'$ , $VT$ contains the $n$ by $n$ orthogonal matrix $V^T$ .		
If $\text{JOB}VT = 'S'$ , $VT$ contains the first $\min(m, n)$ rows of $V^T$ (the right singular vectors, stored row-wise).		
If $\text{JOB}VT = 'N'$ or ' $O$ ', $VT$ is not referenced.		
11:	$LDVT$ – INTEGER	<i>Input</i>
<i>On entry:</i> the first dimension of the array $VT$ as declared in the (sub)program from which F08KBF (DGESVD) is called.		
<i>Constraints:</i>		
if $\text{JOB}VT = 'A'$ , $LDVT \geq \max(1, N)$ ; if $\text{JOB}VT = 'S'$ , $LDVT \geq \max(1, \min(M, N))$ ; otherwise $LDVT \geq 1$ .		
12:	$WORK(\max(1, LWORK))$ – REAL (KIND=nag_wp) array	<i>Workspace</i>
<i>On exit:</i> if $\text{INFO} = 0$ , $WORK(1)$ returns the optimal $LWORK$ .		
If $\text{INFO} > 0$ , $WORK(2 : \min(M, N))$ contains the unconverged superdiagonal elements of an upper bidiagonal matrix $B$ whose diagonal is in $S$ (not necessarily sorted). $B$ satisfies $A = UV^T$ , so it has the same singular values as $A$ , and singular vectors related by $U$ and $V^T$ .		
13:	$LWORK$ – INTEGER	<i>Input</i>
<i>On entry:</i> the dimension of the array $WORK$ as declared in the (sub)program from which F08KBF (DGESVD) is called.		
If $LWORK = -1$ , a workspace query is assumed; the routine only calculates the optimal size of the $WORK$ array, returns this value as the first entry of the $WORK$ array, and no error message related to $LWORK$ is issued.		
<i>Suggested value:</i> for optimal performance, $LWORK$ should generally be larger. Consider increasing $LWORK$ by at least $nb \times \min(M, N)$ , where $nb$ is the optimal <b>block size</b> .		
<i>Constraint:</i> $LWORK \geq \max(1, 3 \times \min(M, N) + \max(M, N), 5 \times \min(M, N))$ .		

14: INFO – INTEGER

Output

*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO &lt; 0

If INFO =  $-i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO &gt; 0

If F08KBF (DGESVD) did not converge, INFO specifies how many superdiagonals of an intermediate bidiagonal form did not converge to zero.

## 7 Accuracy

The computed singular value decomposition is nearly the exact singular value decomposition for a nearby matrix  $(A + E)$ , where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and  $\epsilon$  is the *machine precision*. In addition, the computed singular vectors are nearly orthogonal to working precision. See Section 4.9 of Anderson *et al.* (1999) for further details.

## 8 Parallelism and Performance

F08KBF (DGESVD) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08KBF (DGESVD) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The total number of floating-point operations is approximately proportional to  $mn^2$  when  $m > n$  and  $m^2n$  otherwise.

The singular values are returned in descending order.

The complex analogue of this routine is F08KPF (ZGESVD).

## 10 Example

This example finds the singular values and left and right singular vectors of the 6 by 4 matrix

$$A = \begin{pmatrix} 2.27 & -1.54 & 1.15 & -1.94 \\ 0.28 & -1.67 & 0.94 & -0.78 \\ -0.48 & -3.09 & 0.99 & -0.21 \\ 1.07 & 1.22 & 0.79 & 0.63 \\ -2.35 & 2.93 & -1.45 & 2.30 \\ 0.62 & -7.39 & 1.03 & -2.57 \end{pmatrix},$$

together with approximate error bounds for the computed singular values and vectors.

The example program for F08KDF (DGESDD) illustrates finding a singular value decomposition for the case  $m \leq n$ .

## 10.1 Program Text

```

Program f08kbfe

!     F08KBF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
Use nag_library, Only: dgesvd, nag_wp
!     .. Implicit None Statement ..
Implicit None
!     .. Parameters ..
Integer, Parameter :: nb = 64, nin = 5, nout = 6,             &
                     prerr = 0
!     .. Local Scalars ..
Integer :: i, info, lda, ldu, ldvt, lwork, m, n
!     .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: a(:, :), a_copy(:, :), b(:, :), s(:),      &
                                   u(:, :), vt(:, :), work(:)
Real (Kind=nag_wp) :: dummy(1,1)
!     .. Intrinsic Procedures ..
Intrinsic :: max, min, nint
!     .. Executable Statements ..
Write (nout,*) 'F08KBF Example Program Results'
Write (nout,*)
!     Skip heading in data file
Read (nin,*) 
Read (nin,*) m, n
lda = m
ldu = m
ldvt = n
Allocate (a(lda,n),a_copy(m,n),s(n),vt(ldvt,n),u(ldu,m),b(m))

!     Read the m by n matrix A from data file
Read (nin,*)(a(i,1:n),i=1,m)

!     Read the right hand side of the linear system
Read (nin,*) b(1:m)

a_copy(1:m,1:n) = a(1:m,1:n)

!     Use routine workspace query to get optimal workspace.
lwork = -1
!     The NAG name equivalent of dgesvd is f08kbf
Call dgesvd('A','S',m,n,a,lda,s,u,ldu,vt,ldvt,lwork,info)

!     Make sure that there is enough workspace for block size nb.
lwork = max(m+4*n+nb*(m+n),nint(dummy(1,1)))
Allocate (work(lwork))

!     Compute the singular values and left and right singular vectors
!     of A.

!     The NAG name equivalent of dgesvd is f08kbf
Call dgesvd('A','S',m,n,a,lda,s,u,ldu,vt,ldvt,work,lwork,info)

If (info/=0) Then
    Write (nout,99999) 'Failure in DGESVD. INFO =', info
99999  Format (1X,A,I4)
        Go To 100
End If

!     Print the significant singular values of A

Write (nout,*) 'Singular values of A:'
Write (nout,99998) s(1:min(m,n))

```

```

99998 Format (1X,4(3X,F11.4))

If (prerr>0) Then
  Call compute_error_bounds(m,n,s)
End If

If (m>n) Then
!   Compute V*Inv(S)*U^T * b to get least squares solution.
  Call compute_least_squares(m,n,a_copy,m,u,ldu,vt,ldvt,s,b)
End If

100 Continue

Contains
Subroutine compute_least_squares(m,n,a,lda,u,ldu,vt,ldvt,s,b)

!
! .. Use Statements ..
! Use nag_library, Only: dgemv, dnrm2
! .. Implicit None Statement ..
Implicit None
!
! .. Scalar Arguments ..
Integer, Intent (In) :: lda, ldu, ldvt, m, n
!
! .. Array Arguments ..
Real (Kind=nag_wp), Intent (In) :: a(lda,n), s(n), u(ldu,m),
                                    vt(ldvt,n)
Real (Kind=nag_wp), Intent (Inout) :: b(m)
!
! .. Local Scalars ..
Real (Kind=nag_wp) :: alpha, beta, norm
!
! .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: x(:,), y(:,)
!
! .. Intrinsic Procedures ..
Intrinsic :: allocated
!
! .. Executable Statements ..
Allocate (x(n),y(n))

!
! Compute V*Inv(S)*U^T * b to get least squares solution.

!
! y = U^T b
! The NAG name equivalent of dgemv is f06paf
alpha = 1._nag_wp
beta = 0._nag_wp
Call dgemv('T',m,n,alpha,u,ldu,b,1,beta,y,1)

y(1:n) = y(1:n)/s(1:n)

!
! x = V y
Call dgemv('T',n,n,alpha,vt,ldvt,y,1,beta,x,1)

Write (nout,*)
Write (nout,*) 'Least squares solution:'
Write (nout,99999) x(1:n)

!
Find norm of residual ||b-Ax||.
alpha = -1._nag_wp
beta = 1._nag_wp
Call dgemv('N',m,n,alpha,a,lda,x,1,beta,b,1)

norm = dnrm2(m,b,1)

Write (nout,*)
Write (nout,*) 'Norm of Residual:'
Write (nout,99999) norm

If (allocated(x)) Then
  Deallocate (x)
End If
If (allocated(y)) Then
  Deallocate (y)
End If

99999 Format (1X,4(3X,F11.4))

```

```

End Subroutine compute_least_squares

Subroutine compute_error_bounds(m,n,s)

!      Error estimates for singular values and vectors is computed
!      and printed here.

!      .. Use Statements ..
Use nag_library, Only: ddisna, nag_wp, x02ajf
!      .. Implicit None Statement ..
Implicit None
!      .. Scalar Arguments ..
Integer, Intent (In)          :: m, n
!      .. Array Arguments ..
Real (Kind=nag_wp), Intent (In) :: s(n)
!      .. Local Scalars ..
Real (Kind=nag_wp)            :: eps, serrbd
Integer                      :: i, info
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: rcondu(:, ), rcondv(:, ), uerrbd(:, ), &
                                  verrbd(:, )
!      .. Executable Statements ..
Allocate (rcondu(n),rcondv(n),uerrbd(n),verrbd(n))

!      Get the machine precision, EPS and compute the approximate
!      error bound for the computed singular values. Note that for
!      the 2-norm, S(1) = norm(A)

eps = x02ajf()
serrbd = eps*s(1)

!      Call DDISNA (F08FLF) to estimate reciprocal condition
!      numbers for the singular vectors

Call ddisna('Left',m,n,s,rcondu,info)
Call ddisna('Right',m,n,s,rcondv,info)

!      Compute the error estimates for the singular vectors

Do i = 1, n
    uerrbd(i) = serrbd/rcondu(i)
    verrbd(i) = serrbd/rcondv(i)
End Do

!      Print the approximate error bounds for the singular values
!      and vectors

Write (nout,*)
Write (nout,*) 'Error estimate for the singular values'
Write (nout,99999) serrbd
Write (nout,*)
Write (nout,*) 'Error estimates for the left singular vectors'
Write (nout,99999) uerrbd(1:n)
Write (nout,*)
Write (nout,*) 'Error estimates for the right singular vectors'
Write (nout,99999) verrbd(1:n)

99999 Format (4X,1P,6E11.1)

End Subroutine compute_error_bounds

End Program f08kbfe

```

## 10.2 Program Data

F08KBF Example Program Data

```
6      4                      :Values of M and N
2.27  -1.54   1.15  -1.94
0.28  -1.67   0.94  -0.78
-0.48  -3.09   0.99  -0.21
1.07   1.22   0.79   0.63
-2.35   2.93  -1.45   2.30
0.62  -7.39   1.03  -2.57  :End of matrix A
1.0    1.0    1.0    1.0
1.0    1.0                  :RHS b(1:m)
```

## 10.3 Program Results

F08KBF Example Program Results

```
Singular values of A:
  9.9966        3.6831        1.3569        0.5000
Least squares solution:
 -0.0563       -0.1700        0.8202        0.5545
Norm of Residual:
  1.7472
```

---